# Maintainer: Alexander F. Rødseth <xyproto@archlinux.org>
# Contributor: Crestwave <crest.wave@yahoo.com>

pkgname=fff
pkgver=2.2
pkgrel=3
pkgdesc='Simple and fast file manager'
arch=(any)
url='https://github.com/dylanaraps/fff'
license=(MIT)
makedepends=(git)
depends=(xdg-utils)
source=("git+${url}#commit=5045c4ab91f559a3735619dacf604e37632160cf") # tag: 2.2
sha256sums=('SKIP')

package() {
  cd "$pkgname"
  install -Dm755 -t "$pkgdir/usr/bin" $pkgname
  install -Dm644 -t "$pkgdir/usr/share/man/man1" $pkgname.1
  install -Dm644 -t "$pkgdir/usr/share/doc/$pkgname" README.md
  install -Dm644 -t "$pkgdir/usr/share/licenses/$pkgname" LICENSE.md
}

# getver: github.com/dylanaraps/fff/releases/latest
# vim: ts=2 sw=2 et:
